```
'                    Predicted Heat Strain (PHS) model:
'              computation programme written in Quick Basic

'      This programme is the exact copy of the programme
'      in annex D of the ISO 7933 standard.
'      It was developed by Prof. J. Malchaire,
'      Director of the Occupational Hygiene and Work Physiology Unit
'      University of Louvain in Belgium.
'      Clos Chapelle-aux-Champs 30-38, 1200  Bruxelles
'      E-mail: jacques.malchaireclouvain.be
'      Site www.deparisnet.be

' INITIALISATION
       CLS
'      The user must make sure that, at this point in the programme,
'      the following parameters are available.
'      Standard values must be replaced by actual values if necessary.
'     The water replacement is supposed to be sufficient so that the
'      workers can drink freely (DRINK=1), otherwise the value DRINK=0
'      must be used

       Drink = 1
      weight = 75: '  body mass       kilogrammes
       height = 1.8: '  body height     meters
       Adu = .202 * weight ^ .425 * height ^ .725
       spHeat = 57.83 * weight / Adu

       SWp = 0
       SWtot = 0: Tre = 36.8: Tcr = 36.8: Tsk = 34.1: Tcreq = 36.8: TskTcrwg = .3
       Dlimtre = 0: Dlimloss50 = 0: Dlimloss95 = 0
       Dmax50 = .075 * weight * 1000
       Dmax95 = .05 * weight * 1000

' EXPONENTIAL AVERAGING CONSTANTS
' Core temperature as a function of the metabolic rate: time constant: 10 minutes
       ConstTeq = EXP(-1 / 10)
' Skin Temperature: time constant: 3 minutes
       ConstTsk = EXP(-1 / 3)
' Sweat rate: time constant: 10 minutes
       ConstSW = EXP(-1 / 10)

Duration = 480: 'the duration of the work sequence in minutes

FOR time = 1 TO Duration
' INITIALISATION MIN PER MIN
Tsk0 = Tsk: Tre0 = Tre: Tcr0 = Tcr: Tcreq0 = Tcreq: TskTcrwg0 = TskTcrwg

' INPUT OF THE PRIMARY PARAMETERS
' The user must make sure that, at this point in the programme,
' the following parameters are available. In order for the user
' to test rapidly the programme, the data for the first case
' in annex E of the ISO 7933 standard are introduced.

Ta = 40: 'air temperature degrees celsius
Tr = 40: 'mean radiant temperature        degrees celsius
Pa = 2.5: 'partial water vapour pressure   kilopascals
Va = .3:   'air velocity    metres per second
```

```
Met = 150: 'metabolic rate  Watts per square meter
Work = 0: 'effective mechanical power      Watts per square metre
'Posture posture = 1 sitting, =2 standing, =3 crouching
posture = 2
Icl = .5:  'static  thermal insulation             clo
imst = .38: 'static moisture permeability index    dimensionless
Ap = .54: 'fraction of the body surface covered
        'by the reflective clothing             dimensionless
Fr = .97: 'emissivity of the reflective clothing   dimensionless
        '(by default: Fr=0.97)
'Ardu dimensionless
defspeed = 0: 'code =1 if walking speed entered, 0 otherwise
Walksp = 0: 'walking speed   metres per second
defdir = 0: 'code =1 if walking direction entered, 0 otherwise
THETA = 0: 'angle between walking direction and wind direction       degrees
accl = 100: 'code =100  if acclimatised subject, 0 otherwise

' Effective radiating area of the body
      IF posture = 1 THEN Ardu = .7
      IF posture = 2 THEN Ardu = .77
      IF posture = 3 THEN Ardu = .67

' EVALUATION OF THE MAXIMUM SWEAT RATE AS A FUNCTION OF THE METABOLIC RATE
      SWmax = (Met - 32) * Adu
      IF SWmax > 400 THEN SWmax = 400
      IF SWmax < 250 THEN SWmax = 250
' For acclimatised subjects (accl=100), the maximum Sweat Rate is greater by 25%
      IF accl >= 50 THEN SWmax = SWmax * 1.25
      IF accl < 50 THEN Wmax = .85 ELSE Wmax = 1

' EQUILIBRIUM CORE TEMPERATURE ASSOCIATED TO THE METABOLIC RATE
      Tcreqm = .0036 * Met + 36.6
' Core temperature at this minute, by exponential averaging
      Tcreq = Tcreq0 * ConstTeq + Tcreqm * (1 - ConstTeq)
' Heat storage associated with this core temperature increase during the last minute
      dStoreq = spHeat * (Tcreq - Tcreq0) * (1 - TskTcrwg0)

' SKIN TEMPERATURE PREDICTION
' Skin Temperature in equilibrium
' Clothed model
      Tskeqcl = 12.165 + .02017 * Ta + .04361 * Tr + .19354 * Pa - .25315 * Va
      Tskeqcl = Tskeqcl + .005346 * Met + .51274 * Tre
' Nude model
      Tskeqnu = 7.191 + .064 * Ta + .061 * Tr + .198 * Pa - .348 * Va
      Tskeqnu = Tskeqnu + .616 * Tre
' Value at this minute, as a function of the clothing insulation
      IF Icl >= .6 THEN Tskeq = Tskeqcl: GOTO Tsk
      IF Icl <= .2 THEN Tskeq = Tskeqnu: GOTO Tsk
' Interpolation between the values for clothed and nude subjects, if  0.2 < clo < 0.6
      Tskeq = Tskeqnu + 2.5 * (Tskeqcl - Tskeqnu) * (Icl - .2)
' Skin Temperature at this minute, by exponential averaging
Tsk:
      Tsk = Tsk0 * ConstTsk + Tskeq * (1 - ConstTsk)
' Saturated water vapour pressure at the surface of the skin
      Psk = .6105 * EXP(17.27 * Tsk / (Tsk + 237.3))

' CLOTHING INFLUENCE ON EXCHANGE COEFFICIENTS
```

```
' Static clothing insulation
      Iclst = Icl * .155
' Clothing area factor
      fcl = 1 + .3 * Icl
' Static boundary layer thermal insulation in quiet air
      Iast = .111
' Total static insulation
      Itotst = Iclst + Iast / fcl

' Relative velocities due to air velocity and movements
      IF defspeed > 0 THEN
            IF defdir = 1 THEN
            ' Unidirectional walking
                  Var = ABS(Va - Walksp * COS(3.14159 * THETA / 180))
            ELSE
            ' Omni-directional walking
                  IF Va < Walksp THEN Var = Walksp ELSE Var = Va
            END IF
      ELSE
            ' Stationary or undefined speed
            Walksp = .0052 * (Met - 58): IF Walksp > .7 THEN Walksp = .7
            Var = Va
      END IF

' Dynamic clothing insulation
' Clothing insulation correction for wind (Var) and walking (Walksp)
      Vaux = Var: IF Var > 3 THEN Vaux = 3
      Waux = Walksp: IF Walksp > 1.5 THEN Waux = 1.5
      CORcl = 1.044 * EXP((.066 * Vaux - .398) * Vaux + (.094 * Waux - .378) * Waux)
      IF CORcl > 1 THEN CORcl = 1
      CORia = EXP((.047 * Var - .472) * Var + (.117 * Waux - .342) * Waux)
      IF CORia > 1 THEN CORia = 1
      CORtot = CORcl
      IF Icl <= .6 THEN CORtot = ((.6 - Icl) * CORia + Icl * CORcl) / .6

      Itotdyn = Itotst * CORtot
      IAdyn = CORia * Iast
      Icldyn = Itotdyn - IAdyn / fcl

' Permeability index
' Correction for wind and walking
      CORe = (2.6 * CORtot - 6.5) * CORtot + 4.9
      imdyn = imst * CORe: IF imdyn > .9 THEN imdyn = .9
' Dynamic evaporative resistance
      Rtdyn = Itotdyn / imdyn / 16.7

' HEAT EXCHANGES
' Heat exchanges through respiratory convection and evaporation
      ' temperature of the expired air
      Texp = 28.56 + .115 * Ta + .641 * Pa
      Cres = .001516 * Met * (Texp - Ta)
      Eres = .00127 * Met * (59.34 + .53 * Ta - 11.63 * Pa)

'  Mean temperature of the clothing: Tcl
' Dynamic convection coefficient
      Z = 3.5 + 5.2 * Var
      IF Var > 1 THEN Z = 8.7 * Var ^ .6
```

```
	Hcdyn = 2.38 * ABS(Tsk - Ta) ^ .25
	IF Z > Hcdyn THEN Hcdyn = Z

	auxR = 5.67E-08 * Ardu
	FclR = (1 - Ap) * .97 + Ap * Fr
	Tcl = Tr + .1
Tcl:
' Radiation coefficient
	Hr = FclR * auxR * ((Tcl + 273) ^ 4 - (Tr + 273) ^ 4) / (Tcl - Tr)
	Tcl1 = ((fcl * (Hcdyn * Ta + Hr * Tr) + Tsk / Icldyn)) / (fcl * (Hcdyn + Hr) + 1 / Icldyn)

	IF ABS(Tcl - Tcl1) > .001 THEN
		Tcl = (Tcl + Tcl1) / 2
		GOTO Tcl
	END IF

' Convection and Radiation heat exchanges
	Conv = fcl * Hcdyn * (Tcl - Ta)
	Rad = fcl * Hr * (Tcl - Tr)
' Maximum Evaporation Rate
	Emax = (Psk - Pa) / Rtdyn
' Required Evaporation Rate
	Ereq = Met - dStoreq - Work - Cres - Eres - Conv - Rad

' INTERPRETATION
' Required wettedness
	wreq = Ereq / Emax

' Required Sweat Rate
'	If no evaporation required: no sweat rate
	IF Ereq <= 0 THEN Ereq = 0: SWreq = 0: GOTO SWp
'	If evaporation is not possible, sweat rate is maximum
	IF Emax <= 0 THEN Emax = 0: SWreq = SWmax: GOTO SWp
'	If required wettedness greater than 1.7: sweat rate is maximum
	IF wreq >= 1.7 THEN wreq = 1.7: SWreq = SWmax: GOTO SWp
'	Required evaporation efficiency
	Eveff = (1 - wreq ^ 2 / 2)
	IF wreq > 1 THEN Eveff = (2 - wreq) ^ 2 / 2
	SWreq = Ereq / Eveff
	IF SWreq > SWmax THEN SWreq = SWmax

SWp:
' Predicted Sweat Rate, by exponential averaging
	SWp = SWp * ConstSW + SWreq * (1 - ConstSW)
	IF SWp <= 0 THEN Ep = 0: SWp = 0: GOTO Storage

' Predicted Evaporation Rate
	k = Emax / SWp
	wp = 1
	IF k >= .5 THEN wp = -k + SQR(k * k + 2)
	IF wp > Wmax THEN wp = Wmax
	Ep = wp * Emax

' Heat Storage
Storage:
	dStorage = Ereq - Ep + dStoreq
```

```
' PREDICTION OF THE CORE TEMPERATURE
      Tcr1 = Tcr0
TskTcr:
' Skin - Core weighting
      TskTcrwg = .3 - .09 * (Tcr1 - 36.8)
      IF TskTcrwg > .3 THEN TskTcrwg = .3
      IF TskTcrwg < .1 THEN TskTcrwg = .1

      Tcr = dStorage / spHeat + Tsk0 * TskTcrwg0 / 2 - Tsk * TskTcrwg / 2
      Tcr = (Tcr + Tcr0 * (1 - TskTcrwg0 / 2)) / (1 - TskTcrwg / 2)
      IF ABS(Tcr - Tcr1) > .001 THEN
      Tcr1 = (Tcr1 + Tcr) / 2: GOTO TskTcr
END IF

' PREDICTION OF THE RECTAL TEMPERATURE
      Tre = Tre0 + (2 * Tcr - 1.962 * Tre0 - 1.31) / 9
      IF Dlimtre = 0 AND Tre >= 38 THEN Dlimtre = time

' Total water loss rate during the minute (in W m-²)
      SWtot = SWtot + SWp + Eres
      SWtotg = SWtot * 2.67 * Adu / 1.8 / 60
      IF Dlimloss50 = 0 AND SWtotg >= Dmax50 THEN Dlimloss50 = time
      IF Dlimloss95 = 0 AND SWtotg >= Dmax95 THEN Dlimloss95 = time
     IF DRINK =0 then Dlimloss95 = Dlimloss95 * 0.6: Dlimloss50 = Dlimloss95

' End of loop on duration

NEXT time

'Dlim computation
      IF Dlimloss50 = 0 THEN Dlimloss50 = Duration
      IF Dlimloss95 = 0 THEN Dlimloss95 = Duration
      IF Dlimtre = 0 THEN Dlimtre = Duration


PRINT "tre="; Tre
PRINT "SWtotg="; SWtotg
PRINT "Dlimtre="; Dlimtre
PRINT "Dlimloss50="; Dlimloss50
PRINT "Dlimloss95="; Dlimloss95

END
```